
#include <machine/asm.h>


#ifdef STRNLEN
/* LINTSTUB: size_t strnlen(const char *, size_t) */
ENTRY(strnlen)
	adds	r3, r0, #1	/* start of src + NUL */
#ifdef __thumb__
	adds	r2, r0, r1	/* &src[maxlen] */
	mov	ip, r2
#else
	adds	ip, r0, r1	/* &src[maxlen] */
#endif
1:	cmp	r0, ip		/* is this of string? */
	beq	2f		/*   yes it is */
#ifdef __thumb__
	ldrb	r2, [r0]	/* read a byte */
	adds	r0, r0, #1	/* advance to next byte */
#else
	ldrb	r2, [r0], #1	/* read a byte */
#endif
	cmp	r2, #0		/* is it a NUL? */
	bne	1b		/*   no, get next byte */
	subs	r0, r0, r3	/* get difference between start and end */
	RET			/* return */
2:	mov	r0, r1		/* get maxlen */
	RET			/* return */
END(strnlen)
#else /* STRNLEN */
/* LINTSTUB: size_t strlen(const char *) */
ENTRY(strlen)
	adds	r3, r0, #1	/* start of src + NUL */
1:
#ifdef __thumb__
	ldrb	r2, [r0]	/* read a byte */
	adds	r0, r0, #1
#else
	ldrb	r2, [r0], #1	/* read a byte */
#endif
	cmp	r2, #0		/* is it a NUL? */
	bne	1b		/*   no, get next byte */
	subs	r0, r0, r3	/* return difference between start and end */
	RET
END(strlen)
#endif /* !STRNLEN */
